

NAG Library Routine Document

G02BGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G02BGF computes means and standard deviations, sums of squares and cross-products of deviations from means, and Pearson product-moment correlation coefficients for selected variables.

2 Specification

```

SUBROUTINE G02BGF (N, M, X, LDX, NVAR, KVAR, XBAR, STD, SSP, LDSSP, R,      &
                  LDR, IFAIL)
INTEGER              N, M, LDX, NVARS, KVAR(NVARS), LDSSP, LDR, IFAIL
REAL (KIND=nag_wp) X(LDX,M), XBAR(NVARS), STD(NVARS),                &
                  SSP(LDSSP,NVARS), R(LDR,NVARS)

```

3 Description

The input data consist of n observations for each of m variables, given as an array

$$[x_{ij}], \quad i = 1, 2, \dots, n (n \geq 2), j = 1, 2, \dots, m (m \geq 2),$$

where x_{ij} is the i th observation on the j th variable, together with the subset of these variables, v_1, v_2, \dots, v_p , for which information is required.

The quantities calculated are:

(a) Means:

$$\bar{x}_j = \frac{1}{n} \sum_{i=1}^n x_{ij}, \quad j = v_1, v_2, \dots, v_p.$$

(b) Standard deviations:

$$s_j = \sqrt{\frac{1}{n-1} \sum_{i=1}^n (x_{ij} - \bar{x}_j)^2}, \quad j = v_1, v_2, \dots, v_p.$$

(c) Sums of squares and cross-products of deviations from zero:

$$S_{jk} = \sum_{i=1}^n (x_{ij} - \bar{x}_j)(x_{ik} - \bar{x}_k), \quad j, k = v_1, v_2, \dots, v_p.$$

(d) Pearson product-moment correlation coefficients:

$$R_{jk} = \frac{S_{jk}}{\sqrt{S_{jj}S_{kk}}}, \quad j, k = v_1, v_2, \dots, v_p.$$

If S_{jj} or S_{kk} is zero, R_{jk} is set to zero.

4 References

None.

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of observations or cases.
Constraint: $N \geq 2$.
- 2: M – INTEGER *Input*
On entry: m , the number of variables.
Constraint: $M \geq 2$.
- 3: X(LDX, M) – REAL (KIND=nag_wp) array *Input*
On entry: $X(i, j)$ must be set to x_{ij} , the value of the i th observation on the j th variable, for $i = 1, 2, \dots, n$ and $j = 1, 2, \dots, m$.
- 4: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G02BGF is called.
Constraint: $LDX \geq N$.
- 5: NVAR – INTEGER *Input*
On entry: p , the number of variables for which information is required.
Constraint: $2 \leq \text{NVAR} \leq M$.
- 6: KVAR(NVAR) – INTEGER array *Input*
On entry: $\text{KVAR}(j)$ must be set to the column number in X of the j th variable for which information is required, for $j = 1, 2, \dots, p$.
Constraint: $1 \leq \text{KVAR}(j) \leq M$, for $j = 1, 2, \dots, p$.
- 7: XBAR(NVAR) – REAL (KIND=nag_wp) array *Output*
On exit: the mean value, \bar{x}_j , of the variable specified in $\text{KVAR}(j)$, for $j = 1, 2, \dots, p$.
- 8: STD(NVAR) – REAL (KIND=nag_wp) array *Output*
On exit: the standard deviation, s_j , of the variable specified in $\text{KVAR}(j)$, for $j = 1, 2, \dots, p$.
- 9: SSP(LDSSP, NVAR) – REAL (KIND=nag_wp) array *Output*
On exit: $\text{SSP}(j, k)$ is the cross-product of deviations, S_{jk} , for the variables specified in $\text{KVAR}(j)$ and $\text{KVAR}(k)$, for $j = 1, 2, \dots, p$ and $k = 1, 2, \dots, p$.
- 10: LDSSP – INTEGER *Input*
On entry: the first dimension of the array SSP as declared in the (sub)program from which G02BGF is called.
Constraint: $\text{LDSSP} \geq \text{NVAR}$.
- 11: R(LDR, NVAR) – REAL (KIND=nag_wp) array *Output*
On exit: $R(j, k)$ is the product-moment correlation coefficient, R_{jk} , between the variables specified in $\text{KVAR}(j)$ and $\text{KVAR}(k)$, for $j = 1, 2, \dots, p$ and $k = 1, 2, \dots, p$.

12: LDR – INTEGER *Input*

On entry: the first dimension of the array R as declared in the (sub)program from which G02BGF is called.

Constraint: $LDR \geq NVARs$.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 2$.

IFAIL = 2

On entry, $NVARs < 2$,
or $NVARs > M$.

IFAIL = 3

On entry, $LDX < N$,
or $LDSSP < NVARs$,
or $LDR < NVARs$.

IFAIL = 4

On entry, $KVAR(j) < 1$,
or $KVAR(j) > M$ for some $j = 1, 2, \dots, NVARs$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

G02BGF does not use *additional precision* arithmetic for the accumulation of scalar products, so there may be a loss of significant figures for large n .

8 Parallelism and Performance

Not applicable.

9 Further Comments

The time taken by G02BGF depends on n and p .

The routine uses a two pass algorithm.

10 Example

This example reads in a set of data consisting of five observations on each of four variables. The means, standard deviations, sums of squares and cross-products of deviations from means, and Pearson product-moment correlation coefficients for the fourth, first and second variables are then calculated and printed.

10.1 Program Text

```

Program g02bgfe

!      G02BGF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: g02bgf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: i, ifail, ldr, ldssp, ldx, m, n, nvars
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: r(:,,:), ssp(:,,:), std(:), x(:,,:),    &
                                xbar(:)
Integer, Allocatable         :: kvar(:)
!      .. Executable Statements ..
Write (nout,*) 'G02BGF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) n, m, nvars

      ldr = nvars
      ldssp = nvars
      ldx = n
      Allocate (r(ldr,nvars),ssp(ldssp,nvars),std(nvars),x(ldx,m),xbar(nvars), &
               kvar(nvars))

!      Read in data
Read (nin,*)(x(i,1:m),i=1,n)

!      Read in column IDs
Read (nin,*) kvar(1:nvars)

!      Display data
Write (nout,99999) 'Number of variables (columns) =', m
Write (nout,99999) 'Number of cases      (rows)      =', n

```

```

Write (nout,*)
Write (nout,*) 'Data matrix is:-'
Write (nout,*)
Write (nout,99998)(i,i=1,m)
Write (nout,99997)(i,x(i,1:m),i=1,n)
Write (nout,*)

! Calculate summary statistics
ifail = 0
Call g02bgf(n,m,x,ldx,nvars,kvar,xbar,std,ssp,ldssp,r,ldr,ifail)

! Display results
Write (nout,*) 'Variable Mean St. dev.'
Write (nout,99995)(kvar(i),xbar(i),std(i),i=1,nvars)
Write (nout,*)
Write (nout,*) 'Sums of squares and cross-products of deviations'
Write (nout,99998) kvar(1:nvars)
Write (nout,99996)(kvar(i),ssp(i,1:nvars),i=1,nvars)
Write (nout,*)
Write (nout,*) 'Correlation coefficients'
Write (nout,99998) kvar(1:nvars)
Write (nout,99996)(kvar(i),r(i,1:nvars),i=1,nvars)

99999 Format (1X,A,I5)
99998 Format (1X,6I12)
99997 Format (1X,I3,4F12.4)
99996 Format (1X,I3,3F12.4)
99995 Format (1X,I5,2F11.4)
End Program g02bgfe

```

10.2 Program Data

```

G02BGF Example Program Data
5 4 3 :: N, M, NVARs
3.0 3.0 1.0 2.0
6.0 4.0 -1.0 4.0
9.0 0.0 5.0 9.0
12.0 2.0 0.0 0.0
-1.0 5.0 4.0 12.0 :: End of X
4 1 2 :: KVARs

```

10.3 Program Results

G02BGF Example Program Results

```

Number of variables (columns) = 4
Number of cases (rows) = 5

```

Data matrix is:-

	1	2	3	4
1	3.0000	3.0000	1.0000	2.0000
2	6.0000	4.0000	-1.0000	4.0000
3	9.0000	0.0000	5.0000	9.0000
4	12.0000	2.0000	0.0000	0.0000
5	-1.0000	5.0000	4.0000	12.0000

Variable	Mean	St. dev.
4	5.4000	4.9800
1	5.8000	5.0695
2	2.8000	1.9235

Sums of squares and cross-products of deviations

	4	1	2
4	99.2000	-57.6000	6.4000
1	-57.6000	102.8000	-29.2000
2	6.4000	-29.2000	14.8000

Correlation coefficients

	4	1	2
4	1.0000	-0.5704	0.1670
1	-0.5704	1.0000	-0.7486
2	0.1670	-0.7486	1.0000
